

Chemical Kinetics Modeling

Gary D. Knott, Ph.D.
Civilized Software, Inc.
12109 Heritage Park Circle
Silver Spring MD 20906
Tel. (301) 962-3711
email: knott@civilized.com
URL: www.civilized.com

The MLAB advanced mathematical and statistical modeling system is an unparalleled tool for mathematical modeling; take a look at the following example. Only MLAB can solve a kinetics-modeling problem like this so easily! (It's just as easy to solve enzyme kinetics, multiple site binding equilibrium, or any other of a wide variety of problems.)

Below we show an interesting example of MLAB modeling for dimer kinetics. Suppose we have two substances, A and B which bind to form a complex C , and the substance C , in turn, binds with itself to form a dimer D . We thus have:



Suppose further we mix 2 mmoles of A and 3 mmoles of B and measure the concentration in mmoles of both C and D at ten equally-spaced times between 7 and 70 seconds. From this data we wish to estimate the association and dissociation constants k_1 , k_2 , k_3 , and k_4 . We may proceed in MLAB as follows.

First we read in the data consisting of values of $c(t)$ and $d(t)$ given at the common times 7:70!10. Although common times are used here, this is *not* required.

```
*data = (7:70!10) &' read(ddata,10,2)
*type data
```

	time	c	d
1:	7	1.065	0.0058
2:	14	1.383	0.2203
3:	21	0.9793	0.4019

```

4: 28      1.107  0.3638
5: 35      0.7289 0.456
6: 42      0.7236 0.5014
7: 49      0.4674 0.715
8: 56      0.6031 0.4723
9: 63      0.6149 0.7219
10: 70     0.3369 0.7294

```

```

*cdata = data col 1:2
*ddata = data col (1,3)

```

Now we define our kinetic model so that $c(t)$ is the concentration of c in mmoles at time t and $d(t)$ is the concentration of d in mmoles at time t .

```

* fct c't(t)=k1*(a0-c-2*d)*(b0-c-2*d)-k2*c-2*d't(t)
* fct d't(t)=k3*c*c-k4*d
* initial c(0)=0
* initial d(0)=0
* a0=2;b0=3

```

Now we guess the values of k_1 , k_2 , k_3 , and k_4 . We may use the results of equilibrium studies, analyzed by MLAB, to know values for the ratios k_1/k_2 and k_3/k_4 .

```

* k1=.02;k2=.002; k3=.02;k4=.002
* constraints q={k1>0,k2>0,k3>0,k4>0}

```

Now we may curve-fit the two ode-system-defined functions, c and d , to estimate k_1 , k_2 , k_3 , and k_4 .

```

* fit(k1,k2,k3,k4), c to cdata, d to ddata, constraints q
final parameter values
      value          error      dependency  parameter
0.06830069826      0.01853684638      0.5090866122  K1
0.009215472845      0.01307266967      0.5011226583  K2
0.01429036692      0.002818099621      0.7828534111  K3
1.616224063e-19     0.004158477431      0.7620523838  K4
6 iterations
CONVERGED
best weighted sum of squares = 2.222001e-01
weighted root mean square error = 1.178453e-01

```

```
weighted deviation fraction = 1.203378e-01
lagrange multiplier[4] = -3.928493584
```

Now we may draw the results of the curve-fit.

```
* m=integrate(c't,d't,0:100!140)
* draw m col (1,2) color red
* draw m col (1,4) color green lt dashed
* draw cdata pt circle lt none color red
* draw ddata pt circle lt none color green
* bottom title "time in seconds"
* left title "mmoles (C and D)"
*
* oformat = nformat; nformat = "%4.4lf"
* v=strval(stdest[1]); s=strval(k1)+" '25TF'R "+substr(v,7:strlen(v))
* title s at (.6,.8) ffract size .015
* v=strval(stdest[2]); s=strval(k2)+" '25TF'R "+substr(v,7:strlen(v))
* title s at (.6,.75) ffract size .015
* v=strval(stdest[3]); s=strval(k3)+" '25TF'R "+substr(v,7:strlen(v))
* title s at (.6,.7) ffract size .015
* v=strval(stdest[4]); s=strval(k4)+" '25TF'R "+substr(v,7:strlen(v))
* title s at (.6,.65) ffract size .015
* nformat=oformat

* view
```

